F08ASFP (PZGEQRF)

NAG Parallel Library Routine Document

Note: Before using this routine, please read the Users' Note for your implementation to check for implementation-dependent details. You are advised to enclose any calls to NAG Parallel Library routines between calls to Z01AAFP and Z01ABFP.

1 Description

F08ASFP (PZGEQRF) computes the QR factorization of a complex m by n matrix A_s (i.e., $A_s = QR$), where A_s is a submatrix of a larger m_A by n_A matrix A, i.e.,

$$A_s(1:m, 1:n) \equiv A(i_A : i_A + m - 1, j_A : j_A + n - 1).$$

Note: if $i_A = j_A = 1$, $m = m_A$ and $n = n_A$, then $A_s = A$.

The unitary matrix Q is not formed explicitly but is represented as a product of elementary reflectors

$$Q = H_0 H_1 \dots H_{k-1}, \quad \text{where } k = \min(m, n).$$

Each H_ℓ has the form

$$H_\ell = I - \tau_\ell v_\ell v_\ell^H,$$

where τ is a complex scalar, and v is complex vector. No pivoting is performed by F08ASFP (PZGEQRF).

2 Specification

```
SUBROUTINE F08ASFP(M, N, A, IA, JA, IDESCA, TAU, WORK, LWORK, INFO)
ENTRY      PZGEQRF(M, N, A, IA, JA, IDESCA, TAU, WORK, LWORK, INFO)
COMPLEX*16         A(*), TAU(*), WORK(LWORK)
INTEGER            M, N, IA, JA, IDESCA(9), LWORK, INFO
```

The ENTRY statement enables the routine to be called by its ScaLAPACK name.

3 Data Distribution

3.1 Definitions

The following definitions are used in describing the data distribution within this document:

m_p	–	the number of rows in the logical processor grid.
n_p	–	the number of columns in the logical processor grid.
p_r	–	the row grid coordinate of the calling processor.
p_c	–	the column grid coordinate of the calling processor.
M_b^X	–	the blocking factor for the distribution of the rows of a matrix X.
N_b^X	–	the blocking factor for the distribution of the columns of a matrix X.
numroc$(\alpha, b_\ell, q, s, k)$	–	a function which gives the **num**ber of **r**ows **o**r **c**olumns of a distributed matrix owned by the processor with the row or column coordinate q (p_r or p_c), where α is the total number of rows or columns of the matrix, b_ℓ is the blocking factor used (M_b^X or N_b^X), s is the row or column coordinate of the processor that possesses the first row or column of the distributed matrix and k is either n_p or m_p. The Library provides the function Z01CAFP (NUMROC) for the evaluation of this function.
indxg2p(i_g, b_ℓ, q, s, k)	–	a function which gives the processor row or column coordinate which possess the row or column index i_g of the distributed full matrix A. The arguments b_ℓ, q, s and k have the same meaning as in the function numroc. The Library provides the function Z01CDFP (INDXG2P) for the evaluation of this function.

3.2 Global and Local Arguments

The input arguments M, N, IA, JA and the array elements IDESCA(1) and IDESCA(3),...,IDESCA(8) are all global and so must have the same values on entry to the routine on every processor. The output argument INFO is global and so will have the same value on exit from the routine on each processor. The remaining arguments are local.

3.3 Distribution Strategy

The matrix A must be partitioned into M_b^A by N_b^A rectangular blocks which are stored in an array A in a cyclic 2-d block distribution. This data distribution is described in more detail in the F08 Chapter Introduction. The array TAU is distributed across the processor columns in a cyclic 2-d block fashion, and is aligned with the rows of the matrix A.

This routine assumes that the data has already been correctly distributed, and if this is not the case will fail to produce correct results. However, the Library provides some utility routines which assist you in distributing data correctly. Descriptions of these routines can be found in Chapters F01 and X04 of the NAG Parallel Library Manual.

4 Arguments

1: M — INTEGER *Global Input*

On entry: the number of rows of the submatrix A_s, m.

Constraint: $0 \leq \mathrm{M} \leq \mathrm{IDESCA}(3)$.

2: N — INTEGER *Global Input*

On entry: the number of columns of the submatrix A_s, n.

Constraint: $0 \leq \mathrm{N} \leq \mathrm{IDESCA}(4)$.

3: A(∗) — COMPLEX*16 array *Local Input/Local Output*

Note: array A is formally defined as a vector. However, you may find it more convenient to consider A as a 2-d array of dimension (IDESCA(9),γ), where
$\gamma \geq$ numroc(JA + N − 1, IDESCA(6), p_c, IDESCA(8), n_p). See the Example Program.

On entry: the local part of the matrix A which may contain parts of the m by n submatrix A_s to be factorized.

On exit: if $m \geq n$, the elements below the diagonal of A_s are overwritten by details of the unitary matrix Q and the upper triangle is overwritten by the corresponding elements of the n by n upper triangular matrix R.
If $m < n$, the strictly lower triangular part of A_s is overwritten by details of the unitary matrix Q and the remaining elements are overwritten by the corresponding elements of the m by n upper trapezoidal matrix R.

4: IA — INTEGER *Global Input*

On entry: the row index of matrix A, i_A, that identifies the first row of the submatrix A_s to be factorized.

Constraint: $1 \leq \mathrm{IA} \leq \mathrm{IDESCA}(3) - \mathrm{M} + 1$.

5: JA — INTEGER *Global Input*

On entry: the column index of matrix A, j_A, that identifies the first column of the submatrix A_s to be factorized.

Constraint: $1 \leq \mathrm{JA} \leq \mathrm{IDESCA}(4) - \mathrm{N} + 1$.

6: IDESCA(9) — INTEGER array *Local Input*

Distribution: the array elements IDESCA(1) and IDESCA(3),...,IDESCA(8) must be global to the processor grid and the elements IDESCA(2) and IDESCA(9) are local to each processor.

On entry: the description array for the matrix A. This array must contain details of the distribution of the matrix A and the logical processor grid.

IDESCA(1), the descriptor type. For this routine, which uses a cyclic 2-d block distribution, IDESCA(1) = 1;
IDESCA(2), the BLACS context (ICNTXT) for the processor grid, usually returned by Z01AAFP;
IDESCA(3), the number of rows, m_A, of the matrix A;
IDESCA(4), the number of columns, n_A, of the matrix A;
IDESCA(5), the blocking factor, M_b^A, used to distribute the rows of the matrix A;
IDESCA(6), the blocking factor, N_b^A, used to distribute the columns of the matrix A;
IDESCA(7), the processor row index over which the first row of the matrix A is distributed;
IDESCA(8), the processor column index over which the first column of the matrix A is distributed;
IDESCA(9), the leading dimension of the conceptual 2-d array A.

Constraints:

IDESCA(1) = 1;
IDESCA(3) $\geq$ 0; IDESCA(4) $\geq$ 0;
IDESCA(5) $\geq$ 1; IDESCA(6) $\geq$ 1;
$0 \leq$ IDESCA(7) $\leq m_p - 1$; $0 \leq$ IDESCA(8) $\leq n_p - 1$;
IDESCA(9) $\geq \max(1,$ numroc(IDESCA(3), IDESCA(5), p_r, IDESCA(7), m_p)).

7: TAU($*$) — COMPLEX*16 array *Local Output*

Note: the dimension of the array TAU must be at least α, where
α = numroc(β, IDESCA(6), p_c, IDESCA(8), n_p) and β = JA + min(M,N) − 1.

On exit: the scalar factors τ_ℓ of the elementary reflectors H_ℓ.

8: WORK(LWORK) — COMPLEX*16 array *Local Workspace/Local Output*

On exit: WORK(1) contains the minimum required value of LWORK.

9: LWORK — INTEGER *Local Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08ASFP (PZGEQRF) is called.

Constraint: LWORK $\geq$ IDESCA(6) $\times$($c_1 + c_2$+ IDESCA(6)), where

c_1 = numroc(M + d_1, IDESCA(5), p_r, d_3, m_p);
c_2 = numroc(N + d_2, IDESCA(6), p_c, d_4, n_p);
d_1 = mod(IA − 1, IDESCA(5));
d_2 = mod(JA − 1, IDESCA(6));
d_3 = indxg2p(IA, IDESCA(5), p_r, IDESCA(7), m_p);
d_4 = indxg2p(JA, IDESCA(6), p_c, IDESCA(8), n_p).

This value of LWORK can be calculated by using the Library function Z01CBFP; i.e.,

LWORK = Z01CBFP(M, N, IA, JA, IDESCA)

10: INFO — INTEGER *Global Output*

On exit: INFO = 0 unless the routine detects an error (see Section 5).

5 Errors and Warnings

If INFO < 0 an explanatory message is output and control returned to the calling program.

INFO < 0

On entry, one of the arguments was invalid:

if the kth argument is a scalar INFO $= -k$;
if the kth argument is an array and its jth element is invalid, INFO $= -(100 \times k + j)$.

This error occurred either because a global argument did not have the same value on all logical processors, or because its value on one or more processors was incorrect. An explanatory message distinguishes between these two cases.

6 Further Comments

The total number of floating-point operations is approximately $\frac{8}{3}n^2(3m-n)$ if $m \geq n$ or $\frac{8}{3}m^2(3n-m)$ if $m < n$. To solve the system of equations $A_sX = B_s$, this routine may be followed by a call to F08ATFP which forms Q explicitly. In terms of the QR factorization, the linear system $A_sX = B_s$ is given by $QRX = B_s$, and hence $RX = Q^HB_s$. If R is triangular then X may be computed using the PBLAS routine PZTRSM. F08AUFP may be used to form Q^HB.

6.1 Algorithmic Detail

For a general m by n matrix A, if $m \geq n$, the factorization is given by:

$$A = Q \begin{pmatrix} R \\ 0 \end{pmatrix}$$

where R is an n by n upper triangular matrix and Q is an m by m unitary matrix. It is sometimes more convenient to write the factorization as

$$A = \begin{pmatrix} Q_1 & Q_2 \end{pmatrix} \begin{pmatrix} R \\ 0 \end{pmatrix}$$

which reduces to

$$A = Q_1R,$$

where Q_1 consists of the first n columns of Q, and Q_2 the remaining $m-n$ columns.

If $m < n$, R is upper trapezoidal, and the factorization can be written

$$A = Q(R_1R_2)$$

where R_1 is an m by m upper triangular matrix and R_2 is an m by n rectangular matrix.

The matrix Q is not formed explicitly but is represented as a product of $\min(m,n)$ elementary reflectors. See [1] for details of the block method used by the routine.

6.2 Parallelism Detail

The Level 3 BLAS operations are carried out in parallel within the routine.

6.3 Accuracy

The computed factorization is the exact factorization of a nearby matrix $A + E$, where

$$\|E\|_2 = \epsilon p(m,n)\|A\|_2,$$

ϵ is ***machine precision*** and $p(m,n)$ is a modest function of m and n.

7 References

[1] Anderson E, Bai Z, Bischof C, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A, Ostrouchov S and Sorensen D (1995) *LAPACK Users' Guide* (2nd Edition) SIAM, Philadelphia

[2] Golub G H and Van Loan C F (1989) *Matrix Computations* Johns Hopkins University Press (2nd Edition), Baltimore

8 Example

To solve the linear least-squares problem

$$\min \|Ax^{(i)} - c^{(i)}\|_2 \quad \text{for} \quad i = 1, 2$$

where $c^{(1)}$ and $c^{(2)}$ are the columns of the matrix C,

$$A = \begin{pmatrix} 0.96 - 0.81i & -0.03 + 0.96i & -0.91 + 2.06i & -0.05 + 0.41i \\ -0.98 + 1.98i & -1.20 + 0.19i & -0.66 + 0.42i & -0.81 + 0.56i \\ 0.62 - 0.46i & 1.01 + 0.02i & 0.63 - 0.17i & -1.11 + 0.60i \\ -0.37 + 0.38i & 0.19 - 0.54i & -0.98 - 0.36i & 0.22 - 0.20i \\ 0.83 + 0.51i & 0.20 + 0.01i & -0.17 - 0.46i & 1.47 + 1.59i \\ 1.08 - 0.28i & 0.20 - 0.12i & -0.07 + 1.23i & 0.26 + 0.26i \end{pmatrix}$$

and

$$C = \begin{pmatrix} -1.54 + 0.76i & 3.17 - 2.09i \\ 0.12 - 1.92i & -6.53 + 4.18i \\ -9.08 - 4.31i & 7.28 + 0.73i \\ 7.49 + 3.65i & 0.91 - 3.97i \\ -5.63 - 2.12i & -5.46 - 1.64i \\ 2.37 + 8.03i & -2.84 - 5.86i \end{pmatrix}.$$

The example uses a 2 by 2 logical processor grid and a 2 by 2 block both A and C.

Note: the listing of the Example Program presented below does not give a full pathname for the data file being opened, but in general the user must give the full pathname in this and any other OPEN statement.

8.1 Example Text

```
*     F08ASFP Example Program Text
*     NAG Parallel Library Release 2. NAG Copyright 1996.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          DT
      PARAMETER        (DT=1)
      INTEGER          MB, NB
      PARAMETER        (MB=2,NB=MB)
      INTEGER          MMAX, NMAX, LDA, RHSMAX, LDC, IAROW, IACOL,
     +                 ICROW, ICCOL, LW, LIWORK
      PARAMETER        (MMAX=6,NMAX=4,LDA=MMAX,RHSMAX=2,LDC=MMAX,
     +                 IAROW=0,IACOL=0,ICROW=0,ICCOL=0,LW=100,
     +                 LIWORK=100)
      COMPLEX*16       ONE
      PARAMETER        (ONE=(1.D+0,0.D+0))
*     .. Local Scalars ..
      INTEGER          IA, IC, ICNTXT, IFAIL, INFO, JA, JC, LWORKE,
     +                 LWORKG, M, MP, N, NP, NRHS
      LOGICAL          ROOT
      CHARACTER*80     FORMAT
*     .. Local Arrays ..
```

```
      COMPLEX*16       A(LDA,NMAX), C(LDC,RHSMAX), TAU(MMAX), WORK(LW)
      INTEGER          IDESCA(9), IDESCC(9)
*     .. External Functions ..
      INTEGER          Z01CBFP, Z01CCFP
      LOGICAL          Z01ACFP
      EXTERNAL         Z01CBFP, Z01CCFP, Z01ACFP
*     .. External Subroutines ..
      EXTERNAL         F08ASFP, F08AUFP, PZTRSM, X04BRFP, X04BSFP,
     +                 Z01AAFP, Z01ABFP
*     .. Intrinsic Functions ..
      INTRINSIC        MIN
*     .. Executable Statements ..
*
      ROOT = Z01ACFP()
      IF (ROOT) WRITE (NOUT,*) 'F08ASFP Example Program Results'
*
      MP = 2
      NP = 2
      IFAIL = 0
*
      CALL Z01AAFP(ICNTXT,MP,NP,IFAIL)
*
      OPEN (NIN,FILE='f08asfpe.d')
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      READ (NIN,*) NRHS
      READ (NIN,*) FORMAT

      IF (M.LE.MMAX .AND. N.LE.NMAX .AND. NRHS.LE.RHSMAX) THEN
*
*         Set array descriptor for A, and read A from data file
*
         IDESCA(1) = DT
         IDESCA(2) = ICNTXT
         IDESCA(3) = M
         IDESCA(4) = N
         IDESCA(5) = MB
         IDESCA(6) = NB
         IDESCA(7) = IAROW
         IDESCA(8) = IACOL
         IDESCA(9) = LDA
*
         IFAIL = 0
         CALL X04BRFP(NIN,M,N,A,1,1,IDESCA,IFAIL)
*
*         Set array descriptor for C, and read C from data file
*
         IDESCC(1) = DT
         IDESCC(2) = ICNTXT
         IDESCC(3) = M
         IDESCC(4) = NRHS
         IDESCC(5) = MB
         IDESCC(6) = NB
         IDESCC(7) = ICROW
         IDESCC(8) = ICCOL
         IDESCC(9) = LDC
*
```

```
         IFAIL = 0
*
         CALL X04BRFP(NIN,M,NRHS,C,1,1,IDESCC,IFAIL)
*
*         Factorize A
*
         IA = 1
         JA = 1
         LWORKE = MIN(LW,Z01CBFP(M,N,IA,JA,IDESCA))
*
         CALL F08ASFP(M,N,A,IA,JA,IDESCA,TAU,WORK,LWORKE,INFO)

         IF (INFO.EQ.0) THEN
*
*           Compute solution
*
            IC = 1
            JC = 1
            LWORKG = MIN(LW,Z01CCFP('L',M,N,IA,JA,IC,JC,IDESCA,IDESCC))
*
*             Apply Q to the right hand sides
*
            CALL F08AUFP('L','C',M,NRHS,N,A,IA,JA,IDESCA,TAU,C,IC,JC,
     +                   IDESCC,WORK,LWORKG,INFO)
            IF (INFO.EQ.0) THEN
*
*                     Solve the triangular system
*
               CALL PZTRSM('Left','Upper','No transpose','Non-unit',N,
     +                     NRHS,ONE,A,IA,JA,IDESCA,C,IC,JC,IDESCC)
*
               IF (ROOT) THEN
                  WRITE (NOUT,*)
                  WRITE (NOUT,*) 'Least-squares solution(s).'
                  WRITE (NOUT,*)
               END IF

               IFAIL = 0
               CALL X04BSFP(NOUT,N,NRHS,C,IC,JC,IDESCC,FORMAT,WORK,
     +                      IFAIL)
            END IF
         END IF
*
      END IF
*
      CLOSE (NIN)
*
      IFAIL = 0
      CALL Z01ABFP(ICNTXT,'N',IFAIL)
*
      STOP
      END
```

8.2 Example Data

```
F08ASFP Example Program Data
 6 4                                                          :Values of M and N
 2                                                            :Value of NRHS
 'F7.4'                                                       :Value of FORMAT
 ( 0.96,-0.81) (-0.03, 0.96) (-0.91, 2.06) (-0.05, 0.41)
 (-0.98, 1.98) (-1.20, 0.19) (-0.66, 0.42) (-0.81, 0.56)
 ( 0.62,-0.46) ( 1.01, 0.02) ( 0.63,-0.17) (-1.11, 0.60)
 (-0.37, 0.38) ( 0.19,-0.54) (-0.98,-0.36) ( 0.22,-0.20)
 ( 0.83, 0.51) ( 0.20, 0.01) (-0.17,-0.46) ( 1.47, 1.59)
 ( 1.08,-0.28) ( 0.20,-0.12) (-0.07, 1.23) ( 0.26, 0.26)   :End of matrix A
 (-1.54, 0.76) ( 3.17,-2.09)
 ( 0.12,-1.92) (-6.53, 4.18)
 (-9.08,-4.31) ( 7.28, 0.73)
 ( 7.49, 3.65) ( 0.91,-3.97)
 (-5.63,-2.12) (-5.46,-1.64)
 ( 2.37, 8.03) (-2.84,-5.86)                                 :End of matrix C
```

8.3 Example Results

```
F08ASFP Example Program Results

Least-squares solution(s).

(-0.4936,-1.1993) ( 0.7535, 1.4404)
(-2.4708, 2.8373) ( 5.1726,-3.6235)
( 1.5060,-2.1830) (-2.6609, 2.1334)
( 0.4459, 2.6848) (-2.6966, 0.2711)
```